
NPN_GUM
* MEASURE IS,ISE,NF,NE,RE,IKF,BETA 

*
*                 ________
*                |        | VC
*                |C      _|_
*         B     _|      /2v \
*         ____|'  npnv  \___/
*        |    |`->        |
*    VB _|_      |  0     |
*      /.7v\     |________|
*      \___/      _|_
*        |        ///
*       _|_
*       ///

VC       C  0  DC  2V
VB       B  0  0V
Q1       C  B  0   NPNV

.OPTIONS  GMIN=1e-15 METHOD=gear ABSTOL=1e-15 

.control
dc vb  .1V 1.4V .1V    vc 2V  5V 3V
plot mag(-i(vc)) mag(-i(vb))  vs mag(V(b))      ylog   title   Gummel
plot mag(-i(vc)) mag(-i(vb))                    ylog   title   Gummel
plot mag(i(vc)/i(vb))         vs mag(vc#branch) loglog title   Beta_vs_IC 
plot mag(i(vc)/i(vb))         vs mag(v(b))      ylog   title   Beta_vs_Vb  
.endc

.MODEL NPNV     NPN(
+ IS=15.51E-18  NF=1.005     BF=110      VAF=130.2    IKF=0.0057  
+               NR=1.006     BR=0.4822   VAR=4.286    IKR=0.0002472
+ ISE=9.15E-17  NE=2                                 
+ ISC=1E-21     NC=2                                                 
+ RB=732        RBM=441.2                             IRB=7.5E-04 
+ RE=15.33      RC=109.1                                         
+ CJE=1.727E-14 VJE=0.6408   MJE=0.2563                           
+ CJC=1.826E-14 VJC=0.6399   MJC=0.3531                         
+ CJS=2.939E-14 VJS=0.3488   MJS=0.1813  XCJC=0.4201                 
+ TF=1.65E-11   XTF=1.25     VTF=1       ITF=0.003532         
+ TR=6E-09      FC=0.88      PTF=205                                 
+ KF=1.000E-16  AF=1      
+ XTB=2         EG=1.11      XTI=5       TNOM=25       )

.END



**                                ___
**                               |(C)|    NPN
**     SPICE MODEL               |___|
**    <actual_behavior>            |
**    <varies from spice model !>  / RC 110ohms    VAF=215
**                                 \
**                                 /                    4fF
**         ________________________|_____________________
**        |       |      |      |    |      |     _|_   _|_Cjs
**        |       |12fF  |      |    |Ir/BR |Irn / _ \  ___      
**        |       |      |      /   _|_    _|_   \/ \/   |gnd!
**       _|_Cjcx _|_Cjc _|_Cdc  \    ^      ^    /\_/\  _|_
**       ___     ___    ___     /   /_\    /_\   \___/ \sub/
**   ___  |   Rbb'|      | gmin \    |      |   Ic |    \ /
**  |(B)|_|_/\  __|______|______|____|______|      | |   V
**  |___|     \/  |      |      |   _|_    _|_     | V (hidden)
**      600ohms  _|_Cje _|_Cde  \   \ /    \ /     |
**     BF=116    ___    ___     /   _v_    _v_     |
**                |20fF  | gmin \    |If/Bf |Ifn   |
**                |______|______|____|______|______|
**                                 |
**                  TF=8ps         / RE 45ohms
**                                 \
**                                 /
**                                _|_           
**                               |(E)|
**                               |___|
**            
**




